TD3 - Intégrales et séries

Nicolas Estibals*

26 novembre, 3 et 10 décembre 2008

Exercice 1 Convergence de série

Le but de cet exercice est d'étudier la convergence de la série $\sum_{n>0} a_n$ où a_n est défini par :

$$a_n = \int_0^1 1 - \sqrt[n]{1-t^n}\, dt.$$

1. On pose $f_n : t \mapsto 1 - \sqrt[n]{1-t^n}$, tracez le graphe des fonctions f_1 à f_{10}.
2. Montrez que f_n est croissante pour tout n.
3. Déterminez α_n tel que $f_n(t) \leq 1/n^2$ pour $t \in [0, \alpha_n]$.
4. En déduire un majorant de $\int_0^{\alpha_n} f_n(t)\, dt$.
5. De même, calculez un majorant de $\int_{\alpha_n}^{1-1/n^2} f_n(t)\, dt$ et de $\int_{1-1/n^2}^{1} f_n(t)\, dt$.
6. Conclure sur la convergence de la série $\sum_{n>0} a_n$.
   Indication : estimez asymptotiquement un majorant des a_n.

Exercice 2 Une intégrale généralisé

1. Calculez $\int_0^1 \frac{\ln x}{\sqrt{1-x}}\, dx$ à l'aide de Maple.
2. Tracez $x \mapsto \frac{\ln x}{\sqrt{1-x}}$ et justifiez que cette intégrale pose problème.
   Maple est capable de calculer des intégrales mais ne justifie pas leur convergence il est donc nécessaire de la vérifier
3. Vous avez dû constater que cette intégrale est impropre en 0. Justifiez la convergence de $\int_0^{1/2} \frac{\ln x}{\sqrt{1-x}}\, dx$
4. Vous avez aussi dû constater qu'elle est impropre en 1. Justifiez la convergence de $\int_{1/2}^{1} \frac{\ln x}{\sqrt{1-x}}\, dx$
5. Conclure.

Exercice 3 Calculons π

1. Calculez le développement en série entière de arcsin.
2. Trouvez une formule contenant π et arcsin.
3. Donnez une estimation de π.
4. Bornez l'erreur réalisée, reconsidérez éventuellement la justesse de vos calculs.

*N'hésitez pas à m'écrire à l'adresse `Nicolas.Estibals@ens-lyon.fr`!
Vous pouvez retrouver les sujets sur `http://perso.ens-lyon.fr/nicolas.estibals/`.

Exercice 1 : Convergence de série

```
> restart;
```

Question 1/

```
> with(plots): # car nous allons tracer des courbes...
  f:=(n,t)->1-(1-t^n)^(1/n); # Donons un nom à notre intégrande cela ser
  plus partique.
  plot({f(n,t)$n=1..10},t=0..1,title="Tracé des 10 premières intégrandes
  sur le domaine d'intégration");
```

Warning, the name changecoords has been redefined

$$f := (n, t) \to 1 - (1 - t^n)^{\left(\frac{1}{n}\right)}$$

Tracé des 10 premières intégrandes sur le domaine d'intégration



Question 2/

Il nous suffit de montrer que la dérivé de f_n est positive sur [0,1].

```
> simplify(diff(f(n,t),t));
```

$$(1 - t^n)^{\left(-\frac{n-1}{n}\right)}\, t^{(n-1)}$$

pour 0<=t<=1, on constate facilement que cette expression est positive.

Question 3/

Comme f est croissante, il suffit de trouver alpha_n tel que f(alpha_n)=1/n^2

```
> alpha(n):=solve(f(n,t)=1/n^2,t);
```

$$\alpha(n) := \mathbf{e}^{\left(\frac{\ln\left(-\mathbf{e}^{\left(\ln\left(\frac{n^2-1}{n^2}\right) n\right)} + 1\right)}{n}\right)}$$

Question 4/

Il suffit de majorer f_n par 1/n^2

```
> Int(f(n,t),t=0..alpha(n))<=Int(1/n^2,t=0..alpha(n));
```

$$Int\left(1 - (1 - t^n)^{\left(\frac{1}{n}\right)}, t = 0 .. \mathbf{e}^{\left(\frac{\ln\left(-\mathbf{e}^{\left(\ln\left(\frac{n^2-1}{n^2}\right) n\right)} + 1\right)}{n}\right)}\right) \leq Int\left(\frac{1}{n^2}, t = 0 .. \mathbf{e}^{\left(\frac{\ln\left(-\mathbf{e}^{\left(\ln\left(\frac{n^2-1}{n^2}\right) n\right)} + 1\right)}{n}\right)}\right)$$

```
> # Posons m1 ce majorant
  m1(n):= alpha(n)/n^2;
```

$$m1(n) := \frac{\mathbf{e}^{\left(\frac{\ln\left(-\mathbf{e}^{\left(\ln\left(\frac{n^2-1}{n^2}\right) n\right)} + 1\right)}{n}\right)}}{n^2}$$

Question 5/

Il va encore s'agir de majorer f_n par son maximum sur l'intervalle choisi.

```
> m2(n):=((1-1/n^2)-alpha(n))*f(n,1-1/n^2);
```

$$m2(n) := \left(1 - \frac{1}{n^2} - e^{\left(\frac{\ln\left(-e^{\left(\ln\left(\frac{n^2-1}{n^2}\right)n\right)}+1\right)}{n}\right)}\right)\left(1 - \left(1 - \left(1 - \frac{1}{n^2}\right)^n\right)^{\left(\frac{1}{n}\right)}\right)$$

```
> Int(f(n,t),t=alpha(n)..1-1/n^2)<=m2(n);
```

$$Int\left(1 - (1 - t^n)^{\left(\frac{1}{n}\right)}, t = e^{\left(\frac{\ln\left(-e^{\left(\ln\left(\frac{n^2-1}{n^2}\right)n\right)}+1\right)}{n}\right)} .. 1 - \frac{1}{n^2}\right) \leq \left(1 - \frac{1}{n^2} - e^{\left(\frac{\ln\left(-e^{\left(\ln\left(\frac{n^2-1}{n^2}\right)n\right)}+1\right)}{n}\right)}\right)\left(1 - \left(1 - \left(1 - \frac{1}{n^2}\right)^n\right)^{\left(\frac{1}{n}\right)}\right)$$

```
> m3(n):=1/n^2*1;
```

$$m3(n) := \frac{1}{n^2}$$

```
> Int(f(n,t),t=1-1/n^2..1)<m3(n);
```

$$\int_{1-\frac{1}{n^2}}^{1} 1 - (1 - t^n)^{\left(\frac{1}{n}\right)} \, dt < \frac{1}{n^2}$$

Question 6/
Nous allons majorer a_n par une série convergente et ainsi montrer que la série converge.

```
> # On utilise en premier lieu la relation de Chasles.
  Int(f(n,t),t=0..1)<Int(f(n,t),t=0..alpha(n))+Int(f(n,t),t=alpha(n)..1-
  /n^2)+Int(f(n,t),t=1-1/n^2..1);
```

$$\int_0^1 1 - (1 - t^n)^{\left(\frac{1}{n}\right)} \, dt < \int_0^{e^{\left(\frac{\ln\left(-e^{\left(\ln\left(\frac{n^2-1}{n^2}\right)n\right)}+1\right)}{n}\right)}} 1 - (1 - t^n)^{\left(\frac{1}{n}\right)} \, dt + \int_{e^{\left(\frac{\ln\left(-e^{\left(\ln\left(\frac{n^2-1}{n^2}\right)n\right)}+1\right)}{n}\right)}}^{1-\frac{1}{n^2}} 1 - (1 - t^n)^{\left(\frac{1}{n}\right)} \, dt + \int_{1-\frac{1}{n^2}}^{1} 1 - (1 - t^n)^{\left(\frac{1}{n}\right)} \, d$$

```
> m(n):=simplify(m1(n)+m2(n)+m3(n));
```

$$m(n) := -\frac{1}{n^2}\left(-2\left(-\left(\frac{n^2-1}{n^2}\right)^n + 1\right)^{\left(\frac{1}{n}\right)} - n^2 + 2\left(-\left(\frac{n^2-1}{n^2}\right)^n + 1\right)^{\left(\frac{1}{n}\right)} n^2 - \left(-\left(\frac{n^2-1}{n^2}\right)^n + 1\right)^{\left(\frac{2}{n}\right)} n^2\right)$$

```
> asympt(m(n),n,3);
```

$$\frac{2 + \ln(n)^2}{n^2} + O\left(\frac{1}{n^3}\right)$$

Comme toutes séries de la forme :

```
> Sum(1/n^(1+epsilon),n);
```

$$\sum_n \frac{1}{n^{(1+\varepsilon)}}$$

avec epsilon strictement positif converge et que m(n) est négligable devant :

```
> 1/n^(3/2);
```

$$\frac{1}{n^{(3/2)}}$$

en effet :

```
> limit((2+(ln(n))^2)/n^2*(1/n^(3/2))^(-1),n=infinity);
```

$$0$$

La série converge donc absolument.

Exercice 2 : Intégrale généralisé

```
> restart;
```

Question 1/

La commande 'int' permet de calculer une intégrale et 'Int' de l'afficher.

```
> Int(ln(x)/sqrt(1-x),x=0..1)=int(ln(x)/sqrt(1-x),x=0..1);
```

$$\int_0^1 \frac{\ln(x)}{\sqrt{1-x}}\,dx = -4 + 4\ln(2)$$

Question 2/

```
> with(plot):
  plot(ln(x)/sqrt(1-x),x=0..1,title="Tracé de l'intégrande sur l'interva
  d'intégration");
Error, invalid input: with expects its 1st argument, pname, to be of typ
{package, module}, but received plot
Warning, inserted missing semicolon at end of statement,
...intégration");
```



L'expression n'est pas défini en x=1 et diverge en x=0.

Question 3/

Montrons d'abord que l'intégrande est équivalente à ln en 0.

```
> limit((ln(x)/sqrt(1-x))*(ln(x))^(-1),x=0);
```

$$1$$

Il suffit alors de montrer que ln est intégrable en 0 : (id est l'intégrale suivante converge pour x -> 0)

```
> Int(ln(t),t=x..1/2)=int(ln(t),t=x..1/2);
```

$$\int_x^{\frac{1}{2}} \ln(t)\,dt = -x\ln(x) + x - \frac{1}{2}\ln(2) - \frac{1}{2}$$

```
> limit(int(ln(t),t=x..1/2),x=0);
```

$$-\frac{1}{2}\ln(2) - \frac{1}{2}$$

Question 4/
Montrons que l'intégrande se prolonge par continuité en 1, c'est à dire qu'elle a une limite en 1.

```
> limit(ln(x)/sqrt(1-x),x=1);
```

$$0$$

Remarque cet argument ne marche pas en 0 car cela diverge.

```
> limit(ln(x)/sqrt(1-x),x=0);
```

$$-\infty$$

Question 5/

Nous avons ainsi la convergence des intégrales sur]0,1/2] et sur [1/2,1[, nous pouvons ainsi conclure que le résultat renvoyé par maple est correct.

Exercice 3 : Calculons Pi

```
> restart;
```

Question 1/

On peut retrouver le dévellopement en série entière de arsin à partir de sa dérivé et ce dévellopement bien connu :

```
> (1+x)^a=Sum(Product(a-k,k=0..n-1)/n!*x^n,n=0..infinity);
```

$$(1+x)^a = \sum_{n=0}^{\infty} \frac{\left(\prod_{k=0}^{n-1} (a-k)\right) x^n}{n!}$$

Ce dévellopement a pour rayon de convergence 1.

```
> diff(arcsin(x),x);
```

$$\frac{1}{\sqrt{1-x^2}}$$

Appliquons la formule, avant de la simplifier puis de l'intégrer.

```
> Diff(arcsin(x),x)=Sum(Product(-1/2-k,k=0..n-1)/n!*
  (-1)^n*x^(2*n),n=0..infinity);
```

$$\frac{d}{dx}\operatorname{arcsin}(x) = \sum_{n=0}^{\infty} \frac{\left(\prod_{k=0}^{n-1} \left(-\frac{1}{2}-k\right)\right) (-1)^n x^{(2n)}}{n!}$$

```
> Diff(arcsin(x),x)=Sum(1/2^n*Product(2*k+1,k=0..n-1)*x^(2*n)/n!,n=0..in
  inity);
```

$$\frac{d}{dx}\operatorname{arcsin}(x) = \sum_{n=0}^{\infty} \frac{\left(\prod_{k=0}^{n-1} (2k+1)\right) x^{(2n)}}{2^n\, n!}$$

```
> arcsin(x)=arcsin(0)+Sum(1/2^n*Product(2*k+1,k=0..n-1)*x^(2*n+1)/(2*n+1
  /n!,n=0..infinity);
```

$$\operatorname{arcsin}(x) = \sum_{n=0}^{\infty} \frac{\left(\prod_{k=0}^{n-1} (2k+1)\right) x^{(2n+1)}}{2^n (2n+1)\, n!}$$

Question 2/

Allons au plus simple :

```
> arcsin(1/2);
```

$$\frac{1}{6}\pi$$

d'où : (les ' ' permettent d'empecher à Maple d'évaluer la comande.)

```
> Pi=6*'arcsin'(1/2);
```

$$\pi = 6 \operatorname{arcsin}\left(\frac{1}{2}\right)$$

Question 3/
Servons-nous du dévellopement en série entière et de la formule précédente. (1/2 est dans le disque de convergence)
Pi est environ :

```
> pi:=6*subs(x=1/2,Sum(1/2^n*Product(2*k+1,k=0..n-1)*x^(2*n+1)/(2*n+1)/n
  ,n=0..50));
```

$$\pi := 6 \left(\sum_{n=0}^{50} \frac{\left(\prod_{k=0}^{n-1} (2k+1)\right) \left(\frac{1}{2}\right)^{(2n+1)}}{2^n (2n+1)\, n!} \right)$$

```
> value(pi);evalf(%);
```

841600888936378945554507051251660064979046883069722391377901746332649 69
8432262597166987706752813267889883169515835956654537533581555014637
51367134826577994393395219886771140983380335472746168 32

$$3.141592654$$

Question 4/

Il s'agit de borné l'erreur faite en négliglant les termes d'ordre supérieur à 50 dans le dévellopement. L'erreur comise est donc :

```
> 6*subs(x=1/2,Sum(1/2^n*Product(2*k+1,k=0..n-1)*x^(2*n+1)/(2*n+1)/n!,n=
  1..infinity));
```

$$6\left(\sum_{n=51}^{\infty}\frac{\left(\prod_{k=0}^{n-1}(2\,k+1)\right)\left(\frac{1}{2}\right)^{(2\,n+1)}}{2^{n}\,(2\,n+1)\,n!}\right)$$

que l'on peut réécrire en:

```
> 6*subs(x=1/2,Sum(1/2*Product((2*k+1)/(2*k+2),k=0..n-1)*x^(2*n+1)/(2*n+
  ),n=51..infinity));
```

$$6\left(\sum_{n=51}^{\infty}\frac{\left(\prod_{k=0}^{n-1}\frac{2\,k+1}{2\,k+2}\right)\left(\frac{1}{2}\right)^{(2\,n+1)}}{2\,(2\,n+1)}\right)$$

et majorer ainsi :

```
> 6*Sum((1/2)^(2*n+1)/(2*n+1),n=51..infinity);
```

$$6\left(\sum_{n=51}^{\infty}\frac{\left(\frac{1}{2}\right)^{(2\,n+1)}}{2\,n+1}\right)$$

```
> value(%);evalf(%);
```

$$\frac{3}{10141204801825835211973625643008}\,\mathrm{LerchPhi}\left(\frac{1}{4},1,\frac{103}{2}\right)$$

$$7.6108112529747910040191042510950143172667301191\,10^{-33}$$

Ainsi nous avons approximé Pi avec à une précision jusqu'au 32 ieme chiffre après la virgule, nous pouvons donc demander à Maple plus de chiffres.

```
> Digits:=32:evalf(value(pi));evalf(Pi);
```

$$3.1415926535897932384626433832795$$

$$3.1415926535897932384626433832795$$

La différence entre notre évalution de pi et celle de Maple est du au erreur de calculs intermédiares.